use "$dataDir/dataset_fires_harvest_daily_weather", clear

		local depvar "shf"
		local cluslvl "id_mills"

*Column 1		
	foreach var in `depvar'{
	reghdfe `var' afterXtreatment temp_celsius rain rain2, abs(id_mills i.year#i.month) vce(cluster `cluslvl')

	estimates store `var'_treat_did_wc
		qui sum `var' if e(sample)==1 & year<2015
		local mean1=r(mean)
		estadd scalar mean1 `mean1' : `var'_treat_did_wc
		estadd local moyearfe "Yes"
		estadd local millfe "Yes"	
		estadd local cluslvl "Mill"
		estadd local wc "Yes"
		estadd local poisson "No"

	}		
			
*Column 2	
	foreach var in `depvar'{
		
	ppmlhdfe `var' afterXtreatment temp_celsius rain rain2, abs(id_mills i.year#i.month) vce(cluster `cluslvl')

		estimates store `var'_treat_poididy_wc
		qui sum `var' if e(sample)==1 & year<2015
		local mean1=r(mean)
		estadd scalar mean1 `mean1' : `var'_treat_poididy_wc
		estadd local moyearfe "Yes"
		estadd local millfe "Yes"	
		estadd local cluslvl "Mill"
		estadd local wc "Yes"
		estadd local poisson "Yes"

	}

	*TABLE DAILY (outcome):
		esttab shf_treat_did_wc shf_treat_poididy_wc using "$mainDir/tables/Tab1_main_fires.tex", replace b(%12.3fc) se(%12.3fc) keep( afterXtreatment) varlabels(afterXtreatment  "After 2015 $\times$ Regulated") label starl(* 0.1 ** 0.05 *** 0.01) stats(mean1 N moyearfe millfe wc cluslvl poisson, fmt(%9.3fc %12.0fc) label("Pre 2015 mean" "Obs." "Month by Year FE" "Mill FE" "Weather controls" "Cluster level" "Poisson")) nonotes